<?xml version="1.0" encoding="utf-8"?>
<xsl:stylesheet version="1.0" xmlns:xsl="http://www.w3.org/1999/XSL/Transform" xmlns:zj="http://www.zju.edu.cn" exclude-result-prefixes="zj">
  <!-- 显示一个空格 -->
  <xsl:template name="zj:WhiteSpace">
    <xsl:text disable-output-escaping="yes">&amp;nbsp;</xsl:text>
  </xsl:template>
  <!-- 将回车替换为<br/>，主要在Detail显示中使用 -->
  <xsl:template name="zj:ProcessText">
    <xsl:param name="Text"/>
    <xsl:choose>
      <xsl:when test="contains($Text,'&#10;')">
        <xsl:value-of select="substring-before($Text,'&#10;')"/>
        <br/>
        <xsl:call-template name="zj:ProcessText">
          <xsl:with-param name="Text" select="substring-after($Text,'&#10;')"/>
        </xsl:call-template>
      </xsl:when>
      <xsl:otherwise>
        <p style="word-break:break-all">
          <xsl:value-of select="$Text"/>
        </p>
      </xsl:otherwise>
    </xsl:choose>
  </xsl:template>
  <xsl:template name="zj:LittleTitle">
    <xsl:param name="Name"/>
    <xsl:param name="Image" select="'../images2/ico_t1.gif'"/>
    <img src="{$Image}" width="23" height="23" class="icotitle" align="absMiddle"/>
    <xsl:value-of select="$Name"/>
  </xsl:template>
  <xsl:template name="zj:ReplaceString">
    <xsl:param name="input"/>
    <xsl:param name="old"/>
    <xsl:param name="new"/>
    <xsl:if test="$input">
      <xsl:choose>
        <xsl:when test="contains($input,$old)">
          <xsl:value-of select="substring-before($input,$old)"/>
          <xsl:value-of select="$new" disable-output-escaping="yes"/>
          <xsl:call-template name="zj:ReplaceString">
            <xsl:with-param name="input" select="substring-after($input,$old)"/>
            <xsl:with-param name="old" select="$old"/>
            <xsl:with-param name="new" select="$new"/>
          </xsl:call-template>
        </xsl:when>
        <xsl:otherwise>
          <xsl:value-of select="$input"/>
        </xsl:otherwise>
      </xsl:choose>
    </xsl:if>
  </xsl:template>
</xsl:stylesheet>
